// File: contracts/FlokiInuNFTRewardPickUp.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.6;

import "@openzeppelin/contracts/access/Ownable.sol";
import "./IMintable.sol";

contract FlokiInuNFTRewardPickUp is Ownable {
    uint256 private _numRewards;
    mapping (uint256 => IMintable) private _nftContracts;
    mapping (uint256 => mapping (address => bool)) private _hasReward;
    mapping (address => bool) public hasClaimed;

    constructor(address[] memory nftContracts_) Ownable() {
        _numRewards = nftContracts_.length;

        for (uint256 i = 0; i < nftContracts_.length; i++) {
            _nftContracts[i] = IMintable(nftContracts_[i]);
        }
    }

    function hasReward(address user) public view returns (bool) {
        for (uint256 i = 0; i < _numRewards; i++) {
            if (_hasReward[i][user]) {
                return true;
            }
        }

        return false;
    }

    function claimReward() external {
        require(hasReward(msg.sender), "FlokiInuNFTRewardPickUp::INELIGIBLE_ADDRESS");
        require(!hasClaimed[msg.sender], "FlokiInuNFTRewardPickUp::ALREADY_CLAIMED");

        hasClaimed[msg.sender] = true;
        for (uint256 i = 0; i < _numRewards; i++) {
            if (_hasReward[i][msg.sender]) {
                _nftContracts[i].mint(msg.sender);
            }
        }
    }

    function addBatch(uint256 reward, address[] memory batch) external onlyOwner {
        require(reward < _numRewards, "FlokiInuNFTRewardPickUp::INVALID_REWARD");

        for (uint256 i = 0; i < batch.length; i++) {
            _hasReward[reward][batch[i]] = true;
        }
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: contracts/IMintable.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.6;

interface IMintable {
    function mint(address to) external;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}